

```

#
# Purpose: Generate Static Sub-Summary Report
# Parameters (In): ID_REPORT; YEAR; STATE
# Parameters (Out): ERROR
# Context:
# Created: 2014-Apr-15 12h15 PST - Vincenzo Menanno
# Notes:
# ---
#
# \\\
Loop
#
#
# DECLARE VARIABLES & PARAMETERS
#
Set Variable [ $_id_report; Value:#P ( "ID_REPORT" ) ]
#
Set Variable [ $_year; Value:Let ( [ Y = #P ( "YEAR" ) ]; If ( IsEmpty ( Y ); "All Years"; Y ) ) ]
Set Variable [ $_state; Value:Let ( [ S = #P ( "STATE" ) ]; If ( IsEmpty ( S ); "All States"; S ) ) ]
#
#
# SEE IF WE HAVE ALREADY RUN THIS REPORT
#
Go to Layout [ "REPORT" (REPORT) ]
Enter Find Mode [ ]
Set Field [ REPORT::YEAR; If ( $_year = "All Years"; "="; "=" & $_year ) ]
Set Field [ REPORT::STATE; If ( $_state = "All States"; "="; "=" & $_state ) ]
New Record/Request
Set Field [ REPORT::ID; $_id_report ]
Omit Record
Set Error Capture [ On ]
Perform Find [ ]
Set Error Capture [ Off ]
If [ Get ( FoundCount ) > 0 ]
    Set Variable [ $_id_report_proxy; Value:REPORT::ID ]
    Enter Find Mode [ ]
    Set Field [ REPORT::ID; $_id_report ]
    Perform Find [ ]
    Set Field [ REPORT::ID_REPORT; $_id_report_proxy ]
    Set Field [ REPORT::STATUS; "Completed" ]
    Commit Records/Requests [ Skip data entry validation; No dialog ]
    Exit Loop If [ True // End single-pass loop ]
End If
#
#
# PROCESSING REPORT RECORD
#
Go to Layout [ "REPORT" (REPORT) ]

```

```

Enter Find Mode [ ]
Set Field [ REPORT::ID; $_id_report ]
Set Error Capture [ On ]
Perform Find [ ]
Set Error Capture [ Off ]
Set Field [ REPORT::STATUS; "Processing" ]
Commit Records/Requests [ Skip data entry validation; No dialog ]
#
#
# ENTER FIND MODE BEFORE GOING TO LAYOUT
#
Enter Find Mode [ ]
Go to Layout [ "BABY_NAME" (BABY_NAME) ]
#
#
# FIND FEMALE VALUES
#
Set Field [ BABY_NAME::YEAR; If ( $_year = "All Years"; ""; $_year ) ]
Set Field [ BABY_NAME::STATE; If ( $_state = "All States"; ""; $_state ) ]
Set Field [ BABY_NAME::GENDER; "F" ]
Set Error Capture [ On ]
Perform Find [ ]
Set Error Capture [ Off ]
Set Variable [ $_tn_female; Value:BABY_NAME::TOTAL_NAMES ]
Set Variable [ $_un_female; Value:BABY_NAME::UNIQUE_NAMES ]
#
#
# FIND MALE VALUES
#
Enter Find Mode [ ]
Set Field [ BABY_NAME::YEAR; If ( $_year = "All Years"; ""; $_year ) ]
Set Field [ BABY_NAME::STATE; If ( $_state = "All States"; ""; $_state ) ]
Set Field [ BABY_NAME::GENDER; "M" ]
Set Error Capture [ On ]
Perform Find [ ]
Set Error Capture [ Off ]
Set Variable [ $_tn_male; Value:BABY_NAME::TOTAL_NAMES ]
Set Variable [ $_un_male; Value:BABY_NAME::UNIQUE_NAMES ]
#
#
# TOTALS AND PERCENTAGES
#
Set Variable [ $_total_names; Value:$_tn_female + $_tn_male ]
Set Variable [ $_unique_names; Value:$_un_female + $_un_male ]
Set Variable [ $_percentage_female; Value:$_un_female / $_unique_names ]
Set Variable [ $_percentage_male; Value:$_un_male / $_unique_names ]
#
#

```

```

# CREATE REPORT LINE RECORDS
#
Go to Layout [ "REPORT ( Sub-Summary Custom )" (REPORT_ITEM) ]
New Record/Request
Set Field [ REPORT_ITEM::ID_REPORT; $_id_report ]
Set Field [ REPORT_ITEM::TYPE; "Row" ]
Set Field [ REPORT_ITEM::GENDER; "F" ]
Set Field [ REPORT_ITEM::TOTAL_NAMES; $_tn_female ]
Set Field [ REPORT_ITEM::UNIQUE_NAMES; $_un_female ]
Set Field [ REPORT_ITEM::PERCENTAGE; $_percentage_female ]
New Record/Request
Set Field [ REPORT_ITEM::ID_REPORT; $_id_report ]
Set Field [ REPORT_ITEM::TYPE; "Row" ]
Set Field [ REPORT_ITEM::GENDER; "M" ]
Set Field [ REPORT_ITEM::TOTAL_NAMES; $_tn_male ]
Set Field [ REPORT_ITEM::UNIQUE_NAMES; $_un_male ]
Set Field [ REPORT_ITEM::PERCENTAGE; $_percentage_male ]
New Record/Request
Set Field [ REPORT_ITEM::ID_REPORT; $_id_report ]
Set Field [ REPORT_ITEM::TYPE; "Total" ]
Set Field [ REPORT_ITEM::TOTAL_NAMES; $_total_names ]
Set Field [ REPORT_ITEM::UNIQUE_NAMES; $_unique_names ]
#
#
# UPDATE REPORT RECORD
#
Go to Layout [ "REPORT" (REPORT) ]
Enter Find Mode [ ]
Set Field [ REPORT::ID; $_id_report ]
Set Error Capture [ On ]
Perform Find [ ]
Set Error Capture [ Off ]
Set Field [ REPORT::STATUS; "Completed" ]
Commit Records/Requests [ Skip data entry validation; No dialog ]
#
#
Exit Loop If [ True // End single-pass loop ]
End Loop
# //////////////////////////////////////
#

```